## Warshall algorithm

Topics: Graph Algorithms

Tools: Why3

see also the index (by topic, by tool, by reference, by year)

Warshall algorithm

Computes the transitive closure of a graph implemented as a Boolean matrix.

Authors: Martin Clochard (École Normale Supérieure) Jean-Christophe Filliâtre (CNRS)

```module WarshallAlgorithm

use import int.Int
use import matrix.Matrix

(* path m i j k =
there is a path from i to j, using only vertices smaller than k *)
inductive path (matrix bool) int int int =
| Path_empty:
forall m: matrix bool, i j k: int.
get m i j -> path m i j k
| Path_cons:
forall m: matrix bool, i x j k: int.
0 <= x < k -> path m i x k -> path m x j k -> path m i j k

lemma weakening:
forall m i j k1 k2. 0 <= k1 <= k2 ->
path m i j k1 -> path m i j k2

lemma decomposition:
forall m k. 0 <= k -> forall i j. path m i j (k+1) ->
(path m i j k \/ (path m i k k /\ path m k j k))

let transitive_closure (m: matrix bool) : matrix bool
requires { m.rows = m.columns }
ensures  { let n = m.rows in
forall x y: int. 0 <= x < n -> 0 <= y < n ->
get result x y <-> path m x y n }
=
let t = copy m in
let n = m.rows in
for k = 0 to n - 1 do
invariant { forall x y. 0 <= x < n -> 0 <= y < n ->
get t x y <-> path m x y k }
for i = 0 to n - 1 do
invariant { forall x y. 0 <= x < n -> 0 <= y < n ->
get t x y <->
if x < i
then path m x y (k+1)
else path m x y k }
for j = 0 to n - 1 do
invariant { forall x y. 0 <= x < n -> 0 <= y < n ->
get t x y <->
if x < i \/ (x = i /\ y < j)
then path m x y (k+1)
else path m x y k }
set t i j (get t i j || get t i k && get t k j)
done
done
done;
t

end
```
{"found_math": false, "script_math_tex": 0, "script_math_asciimath": 0, "math_annotations": 0, "math_alttext": 0, "mathml": 0, "mathjax_tag": 0, "mathjax_inline_tex": 0, "mathjax_display_tex": 0, "mathjax_asciimath": 0, "img_math": 0, "codecogs_latex": 0, "wp_latex": 0, "mimetex.cgi": 0, "/images/math/codecogs": 0, "mathtex.cgi": 0, "katex": 0, "math-container": 0, "wp-katex-eq": 0, "align": 0, "equation": 0, "x-ck12": 0, "texerror": 0}